class IUP_SPIN_BOX
-- An horizontal container that already contains a IUP_SPIN.

inherit
	IUP_CONTAINER
		redefine
			execute_spin
		end
	IUP_WIDGET_CHILD
	IUP_WIDGET_NAME

create {ANY}
	spin_box

feature {ANY}

	spin_box_empty
		-- A new empty spin box.
		local
			a_spin_box, p: POINTER
		do
			a_spin_box := int_spin_box(p)
			set_widget(a_spin_box)
		end

	spin_box(child: IUP_WIDGET)
		-- A new spin box with the given child.
		local
			a_spin_box: POINTER
		do
			a_spin_box := int_spin_box(child.widget)
			set_widget(a_spin_box)
		end

	-- Callback

	set_cb_spin (act: detachable FUNCTION[TUPLE[IUP_SPIN_BOX, INTEGER], STRING])
		-- Called each time the user clicks in the buttons. It will increment 1 
		-- and decrement -1 by default. Holding the Shift key will set a factor 
		-- of 2, holding Ctrl a factor of 10, and both a factor of 100.
		local
			operation: INTEGER
		do
			cb_spin := act

			if cb_spin /= Void then
				operation := 1
			else
				operation := 0
			end
			
			iup_open.set_callback (Current, "SPIN_CB", "NONEEDED", operation)
		end

feature {IUP}
	
	execute_spin (inc: INTEGER): STRING
		do
			if attached cb_spin as int_cb then
				Result := int_cb.item([Current, inc])
			else
				Result := "IUP_DEFAULT"
			end
		end

feature {NONE}

	-- For callback

	cb_spin: detachable FUNCTION[TUPLE[IUP_SPIN_BOX, INTEGER], STRING]

	-- Internal

	int_spin_box (wgt: POINTER): POINTER
		external
			"C inline use %"eiffel-iup.h%""
      alias
			"return IupSpinbox ($wgt);"
      end

end

-- The MIT License (MIT)

-- Copyright (c) 2016, 2019, 2020 by German A. Arias

-- Permission is hereby granted, free of charge, to any person obtaining a copy
-- of this software and associated documentation files (the "Software"), to deal
-- in the Software without restriction, including without limitation the rights
-- to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
-- copies of the Software, and to permit persons to whom the Software is
-- furnished to do so, subject to the following conditions:
--
-- The above copyright notice and this permission notice shall be included in 
-- all copies or substantial portions of the Software.
--
-- THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
-- IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
-- FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
-- AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
-- LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
-- OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
-- SOFTWARE.
